

## **Código do Simulador em ECSL**

Não sendo, no corpo deste trabalho, o modelo de simulação descrito na sua totalidade, é aqui apresentado o código do simulador em ECSL, contendo a descrição de todas as actividades ilustradas no diagrama em ACDs apresentado na Figura 6.3.

```

C#####
C# ECS.RELAY4                               Versao final  #
C#                                           #
C#           CONTROLO DE CONGESTAO EM RELAYS RC51.1111      #
C#                                           #
C# - Controlo dos parâmetros de débito e atraso de trânsito #
C# - Suporte de ligacoes multiplas                #
C# - Com segmentacao e reassemblagem de pacotes    #
C#####

```

```

C*****
C           Seccao de Definicoes

```

```

THERE ARE 5 ChegLan
THERE ARE 5 ChegPsdn

```

```

THERE ARE 5 CanalLan WITH MxPacLan CMDLan NDLan CNDLan
*           IMCLan MxBufLan NBufLan JanOcLan
*           KLAN PacRxLan PacTxPsdn
*           OcuRxLan OcuTxPsdn IndPrLan
*           HopCtLan QoSTrLan QoSDlLan

```

```

THERE ARE 5 CanalPsdn WITH MxPacPsdn CMDPsdn NDPsdn CNDPsdn
*           IMCPsdn MxBufPsdn NBufPsdn JanOcPsdn
*           KPsdn PacRxPsdn PacTxLan
*           OcuRxPsdn OcuTxLan IndPrPsdn
*           HopCtPsdn QoSTrPsdn QoSDlPsdn

```

```

THERE ARE 160 PacLan SET PoolLan WaitLan CanalLan WRelLan
*           CanalLan
*           WITH MbitLan CPLan TInLan TInReLan

```

```

THERE ARE 80 PacPsdn SET PoolPsdn WaitPsdn CanalPsdn WRelPsdn
*           CanalPsdn
*           WITH MbitPsdn CPPsdn TInPsdn TInRePsdn

```

```

THERE ARE 80 Buffer SET BufOcupado BufLivre

```

```

THERE ARE 1 MFisLan SET Livre
THERE ARE 1 TxRxLan
THERE ARE 1 TxPsdn
THERE ARE 1 RxPsdn

```

```

HIST AWaitLan CanalLan (100 0 10) TWaitLan CanalLan (10)
HIST AWRelLan CanalLan (100 0 10) TWRelLan CanalLan (30)
HIST AWaitPsdn CanalPsdn (100 0 10) TWaitPsdn CanalPsdn (30)
HIST AWRelPsdn CanalPsdn (100 0 1) TWRelPsdn CanalPsdn (30)
HIST TBufOcupado (20 0 5)

```

```

FUNCTION PICTURE WEIBULL NEGEXP FIX MEAN MODULO RANDOM MAXOF
REAL FactOcupMeio TaxaLan TaxaPsdn Media
REAL CgLanPsdn CgPsdnLan

```

```

C*****
C      Seccao de Inicializacoes
C
C
C Colocacao das entidades nas filas
C
      PacLan  160 LOAD PoolLan
      PacPsdn 80 LOAD PoolPsdn
      Buffer   80 LOAD BufLivre
      MFisLan  1 LOAD Livre
C
C
C Determinacao do numero de actividades necessario 'a simulacao 'NumAct'
C no caso de este nao ser dado.
C
      CHAIN
      NumAct EQ 0
      AMaxLan = 0
      BMaxLan = 0
      FOR I = Canallan
          AMaxLan = AMaxLan +
*              (CMDLan OF Canallan I * NDlan OF Canallan I)
          IMaxLan = IMCLan OF Canallan I * NDlan OF Canallan I
          BMaxLan = MAXOF (BMaxLan, IMaxLan, 2)
      AMaxLan = FIX (AMaxLan * 8 / TaxaPsdn)
      AMaxPsdn = 0
      BMaxPsdn = 0
      FOR I = CanalPsdn
          AMaxPsdn = AMaxPsdn +
*              (CMDPsdn OF CanalPsdn I * NDPsdn OF CanalPsdn I)
          IMaxPsdn = IMCPsdn OF CanalPsdn I * NDPsdn OF CanalPsdn I
          BMaxPsdn = MAXOF (BMaxPsdn, IMaxPsdn, 2)
      AMaxPsdn = FIX (AMaxPsdn * 8 / TaxaPsdn)
      NumAct = FIX (1.0 * MAXOF (AMaxLan, 1000 * BMaxLan,
*                             AMaxPsdn, 1000 * BMaxPsdn, 4))
      OR CONTINUE
      TYPE 'Numero total de actividades:', NumAct
C
C
C Calculo do intervalo medio entre ocupacoes do meio fisico pelas outras
C estacoes da Lan a partir do factor de ocupacao do meio 'FactOcupMeio'
C
      MediaLiMFis = FIX(((1-FactOcupMeio) / FactOcupMeio) * MediaOcmFis)
C
C
C Calculo das necessidades totais de buffers nos canais da Lan e Psdn.
C O numero total de buffers disponiveis tem que ser >= que o somatorio
C de (K * MaxPac) / SizeBuf para TODOS os canais da LAN e da PSDN para
C garantir que nao ha' perda de informacao por falta de buffers.
C
      NeedBuf = 0
      FOR I = Canallan
          NeedBuf = NeedBuf +
*              (Klan OF Canallan I * MxPacLan OF Canallan I)
      FOR I = CanalPsdn
          NeedBuf = NeedBuf +
*              (KPsdn OF CanalPsdn I * MxPacPsdn OF CanalPsdn I)
C
C
C Inicializacao dos atributos dos canais (JanOcxxx, PacRxxxx, PacTxxxx, CNDxxx
C e MxBufxxx). MxBufxxx e' inicializado de acordo com a seguinte expressao:
C
      MxBufxxx = MAXOF (((QoSdlxxx / HopCtxxx) * QoSTrxxx) / (SizeBuf * 8)
C                      (MxPacxxx OF Canalxxx I / SizeBuf))
C
C
      SomaBuf = 0
      FOR I = Canallan
          MxBufA = 1 + FIX (0.001 * (
*              (QoSdlLan OF Canallan I / HopCtLan OF Canallan I)
*              * QoSTrLan) / (SizeBuf * 8))
          MxBufB = 1 + FIX (1.0 * MxPacLan OF Canallan I / SizeBuf)
          MxBufLan OF Canallan I = MAXOF (MxBufA, MxBufB, 2)

```

```

TYPE MxBufA, MxBufB, MxBufLan OF CanalLan I
SomaBuf = SomaBuf + MxBufLan OF CanalLan I
JanOcLan OF CanalLan I = 0
PacRxLan OF CanalLan I = 0
PacTxPsdn OF CanalLan I = 0
OcuRxLan OF CanalLan I = 0
OcuTxPsdn OF CanalLan I = 0
CNDLan OF CanalLan I = NDlan OF CanalLan I
IndPrLan OF CanalLan I = 1000

```

```

FOR I = CanalPsdn
MxBufA = 1 + FIX (0.001 * (
* (QoSd1Psdn OF CanalPsdn I / HopCtPsdn OF CanalPsdn I)
* * QoSTrPsdn) / (SizeBuf * 8))
MxBufB = 1 + FIX(1.0 * MxPacPsdn OF CanalPsdn I / SizeBuf)
MxBufPsdn OF CanalPsdn I = MAXOF (MxBufA, MxBufB, 2)
TYPE MxBufA, MxBufB, MxBufPsdn OF CanalPsdn I
SomaBuf = SomaBuf + MxBufPsdn OF CanalPsdn I
JanOcPsdn OF CanalPsdn I = 0
PacRxPsdn OF CanalPsdn I = 0
PacTxLan OF CanalPsdn I = 0
OcuRxPsdn OF CanalPsdn I = 0
OcuTxLan OF CanalPsdn I = 0
CNDPsdn OF CanalPsdn I = NDPsdn OF CanalPsdn I
IndPrPsdn OF CanalPsdn I = 1000

```

```

C
C Inicializacao de contadores de pacotes, de apontadores para os
C canais a serem servidos e de outrs variaveis.
C

```

```

ChPacPsdn = 0
ChPacLan = 0

```

```

LastTxRx = 0
PsdnLastCalIndP = 1
LanLastCalIndP = 1

```

```

NBufFalta = 0

```

```

RECYCL

```

```
C*****
C          Seccao de Registos
C
C Alguns registos sao feitos em milisegundos (divisao por mil) para evitar
C overflow no calculo das medias dos histogramas de comprimento das filas
C
    ACTIVITIES NumAct
    Duracao = (CLOCK - PrevCLOCK)
    DurMS = Duracao / 1000

    FOR I = CanalLan
        ADD WaitLan OF CanalLan I TO TWaitLan OF CanalLan I DurMS
        ADD WRelLan OF CanalLan I TO TWRelLan OF CanalLan I DurMS

    FOR I = CanalPsdn
        ADD WaitPsdn OF CanalPsdn I TO TWaitPsdn OF CanalPsdn I DurMS
        ADD WRelPsdn OF CanalPsdn I TO TWRelPsdn OF CanalPsdn I DurMS

    ADD BufOcupado TO TBufOcupado Duracao

    FOR TxRxLan WITH TIME OF TxRxLan LT 0
        ADD Duracao TO ATxRxLan

    PrevCLOCK = CLOCK

    CHAIN
        Tick EQ 1
        EVERY 5000000
        TYPE CLOCK
        OR CONTINUE
```

```
C*****
C      Seccao de Actividades
C
C-----
C      Actividade: Meio Fisico Ocupado (MFisOcup)
C
C Esta actividade simula a ocupacao do meio fisico da Lan pelas outras
C estacoes. O meio fisico e' ocupado com a transmissao de um pacote, durante
C um intervalo de tempo obtido de uma distribuicao de Weibull com media
C 'MediaOcMFis' e alfa = 2 (para simular o efeito da variacao do tamanho dos
C pacotes). As ocupacoes do meio fisico pelas outras estacoes sucedem-se com
C uma cadencia de Poisson de media 'MediaLiMFis', obtida a partir do factor
C de ocupacao do meio fisico 'FactOcuMeio' dado (ver inicializacao).
C
      BEGIN MFisOcupado
      TIME OF TxRxLan LE 0
      MFisLan 1 IN Livre
      Duracao = WEIBULL (FIX (1.13 * MediaOcMFis), 2, Seed)
      TIME OF TxRxLan = Duracao
      AADuracao = Duracao + NEGEXP (MediaLiMFis, Seed)
      MFisLan 1 FROM Livre INTO Livre AFTER AADuracao
      MFisOcupado + 1 AFTER Duracao
      REPEAT
```



```

C-----
C      Actividade: Recepcao PSDN (RecepPsdn)
C
C Esta actividade simula a recepcao pelo Relay de pacotes vindos da PSDN.
C O canal a ser servido e' selecionado aleatoriamente, tem que ter janela
C disponivel e pacotes pendentes, se nao cumprir estas condicoes sao tentados
C sequencialmente, todos os canais. Se o pacote recebido da Psdn for de
C tamanho superior ao tamanho maximo permitido no correspondente canal na
C Lan e' feita a sua segmentacao de modo a ser respeitado esse limite.
C Cada pacote recebido (ou gerado por segmentacao) ocupa um ou mais
C buffers consoante o seu tamanho. A recepcao de um pacote faz avancar a
C janela. E' registado o tempo de chegada dos pacotes ao Relay para,
C aquando da sua transmissao, ser calculado o atraso sofrido.
C Sempre que faltarem buffers para guardar os pacotes recebidos e'
C incrementada a variavel NBufFalta.
C

```

```

BEGIN RecepPsdn
TIME OF RxPsdn LE 0
CanRxPsdn = RANDOM (NCanal, Seed)
FOR J = CanalPsdn
  TIME OF RxPsdn LE 0
  I = MODULO (CanRxPsdn + (J - 1), NCanal) + 1
  JanOcPsdn OF CanalPsdn I LT KPsdn OF CanalPsdn I
  FIND FIRST PacPsdn A IN WaitPsdn OF CanalPsdn I
  ADD FIX(0.001 * (CLOCK - TInPsdn OF PacPsdn A)) TO
*   AWaitPsdn OF CanalPsdn I
  Duracao = FIX ((CPPsdn OF PacPsdn A) * 8 / TaxaPsdn)
  TIME OF RxPsdn = Duracao
  P = 0
  CHAIN
  CPPsdn OF PacPsdn A GT MxPacLan OF Canallan I
  N = CPPsdn OF PacPsdn A / MxPacLan OF Canallan I
  CPPsdn OF PacPsdn A = MODULO (CPPsdn OF PacPsdn A,
*   MxPacLan OF Canallan I)
  FOR 1 TO N
    CHAIN
    FIND FIRST PacPsdn C IN PoolPsdn
    TInRePsdn OF PacPsdn C = CLOCK + Duracao
    MbitPsdn OF PacPsdn C = 1
    CPPsdn OF PacPsdn C = MxPacLan OF Canallan I
    PacPsdn C FROM PoolPsdn INTO WRelPsdn
*   OF CanalPsdn I AFTER Duracao
    P = P + (CPPsdn OF PacPsdn C / SizeBuf)
    CHAIN
    MODULO (CPPsdn OF PacPsdn C, SizeBuf) NE 0
    P = P + 1
    OR CONTINUE
    OR TYPE "Numero insuficiente de pacotes em PoolPsdn"
*   AND FINISH
  OR CONTINUE
  TInRePsdn OF PacPsdn A = CLOCK + Duracao
  OcuRxPsdn OF CanalPsdn I + Duracao AFTER Duracao
  LastTxRx = MAXOF (LastTxRx, CLOCK + Duracao, 2)
  JanOcPsdn OF CanalPsdn I + 1 AFTER Duracao
  PacRxPsdn OF CanalPsdn I + 1 AFTER Duracao
  PacPsdn A FROM WaitPsdn OF CanalPsdn I INTO WRelPsdn
*   OF CanalPsdn I AFTER Duracao
  N = CPPsdn OF PacPsdn A / SizeBuf
  CHAIN
  MODULO (CPPsdn OF PacPsdn A, SizeBuf) NE 0
  N = N + 1
  OR CONTINUE
  FOR 1 TO (N + P)
    CHAIN
    FIND FIRST Buffer B IN BufLivre
    Buffer B FROM BufLivre INTO BufOcupado
    NBufPsdn OF CanalPsdn I + 1
    OR NBufFalta = NBufFalta + 1
  RecepPsdn + 1 AFTER Duracao

```



```

C-----
C      Actividade: Transmissao LAN (TransLan)
C
C Esta actividade simula a transmissao pelo Relay de pacotes nos canais
C da Lan. Os pacotes sao reassemblados sempre que a sequencia Mbit dos
C pacotes pendentes o permita. A reassemblagem e' feita respeitando o
C limite fixado para o tamanho maximo de pacote, de cada canal. Os canais
C sao servidos sequencialmente a contar do ultimo canal servido. E'
C respeitado o efeito de "padding" existente nas Lans CSMA/CD; se o pacote
C a transmitir tiver menos que 'MinFraLan' bytes e' feito o seu enchimento
C ate' se atingir aquele valor (esta operacao so' tem efeito no tempo de
C transmissao dos pacotes). Os canais sao servidos de acordo com o seu
C indice de Partilha (IndPart). Depois de transmitido o pacote os indices
C de Partilha de todos os canais sao actualizados.
C

```

```

BEGIN TransLan
TIME OF TxRxLan LE 0
MinIndPart = 9999999
Canal = 1
FOR I = CanalPsdn
  CHAIN
    EXISTS PacPsdn IN WRelPsdn OF CanalPsdn I
    IndPrPsdn OF CanalPsdn I LE MinIndPart
    MinIndPart = IndPrPsdn OF CanalPsdn I
    Canal = I
    OR CONTINUE
  I = Canal
  EXISTS PacPsdn IN WRelPsdn OF CanalPsdn I
  PacTx = 0
  L = 0
  Falhou = 0
  FOR PacPsdn IN WRelPsdn OF CanalPsdn I
    CHAIN
      Falhou EQ 0
      MbitPsdn OF PacPsdn = 1
      L + CPPsdn OF PacPsdn LE MxPacLan OF CanalLan I
      PacTx = PacTx + 1
      L = L + CPPsdn OF PacPsdn
      OR Falhou = 1
    CHAIN
      L LT MinFraLan
      L = MinFraLan
      OR CONTINUE
    Duracao = FIX (L * 8 / TaxaLan)
    TIME OF TxRxLan = Duracao
    OcuTxLan OF CanalPsdn I + Duracao AFTER Duracao
    LastTxRx = MAXOF (LastTxRx, CLOCK + Duracao, 2)
    PacTxLan OF CanalPsdn I + 1 AFTER Duracao
    FOR 1 TO PacTx
      FIND FIRST PacPsdn A IN WRelPsdn OF CanalPsdn I
      ADD CLOCK - TInRePsdn OF PacPsdn A TO AWRelPsdn OF
*          CanalPsdn I
      PacPsdn A FROM WRelPsdn OF CanalPsdn I INTO PoolPs
*          AFTER Duracao
      N = CPPsdn OF PacPsdn A / SizeBuf
      CHAIN
        MODULO (CPPsdn OF PacPsdn A, SizeBuf) NE 0
        N = N + 1
        OR CONTINUE
      FOR 1 TO N
        NBufPsdn OF CanalPsdn I GT 0
        FIND FIRST Buffer B IN BufOcupado
        Buffer B FROM BufOcupado INTO BufLivre AFTER Duracao
        NBufPsdn OF CanalPsdn I - 1
        TransLan + 1 AFTER Duracao
      IndPrPsdn OF CanalPsdn I = FIX (1000.0 *
*   ( ( (PsdnLastCalIndP * 0.001 * IndPrPsdn OF CanalPsdn I) +
*     (L * 8.0 / (0.000001 * QoSTrPsdn OF CanalPsdn I)) ) /
*   (1. * (CLOCK + Duracao)) ) )
    FOR J = CanalPsdn

```

```
CHAIN
  J NE I
    IndPrPsdn OF CanalPsdn J = FIX (1000 * (PsdnLastCalIndP *
*      0.001 * IndPrPsdn OF CanalPsdn J) /
*      (1. * (CLOCK + Duracao)))
  OR CONTINUE
PsdnLastCalIndP = CLOCK + Duracao
```

```
C-----  
C      Actividade: Confirmacao PSDN (ConfPsdn)  
C  
C Esta actividade simula o envio de confirmacoes pelo Relay aos emissores  
C de pacotes na Psdn. As confirmacoes sao simuladas pelo destravamento da  
C janela dos canais. O destravamento so' e' feito se existir espaco para  
C a recepcao de pelo menos um pacote nos buffers reservados a um canal.  
C O destravamento e' feito com um atraso composto de uma parte constante  
C e de uma parte obtida de uma cadencia de Poisson para simular o atraso  
C na transmissao e propagacao das confirmacoes. As confirmacoes sao feitas  
C rotativamente em todos os canais da Psdn.  
C
```

```
BEGIN ConfPsdn  
FOR I = CanalPsdn  
  JanOcPsdn OF CanalPsdn I GT 0  
  BExist = MxBufPsdn OF CanalPsdn I - NBufPsdn OF CanalPsdn I  
  BPac = MxPacPsdn OF CanalPsdn I / SizeBuf  
  BRes = (KPsdn OF CanalPsdn I - JanOcPsdn OF CanalPsdn I) * BPac  
  BLivre = BExist - BRes  
  J = BLivre / BPac  
  CHAIN  
    J GT 0  
    Duracao = TConfPsdn + NEGEXP (TVarPsdnConf, Seed)  
    JanOcPsdn OF CanalPsdn I + (- J) AFTER Duracao  
    ConfPsdn + J AFTER Duracao
```



```

C-----
C      Actividade: Recepcao LAN (RecepLan)
C
C Esta actividade simula a recepcao pelo Relay de pacotes vindos da LAN.
C O canal a ser servido e' selecionado aleatoriamente, tem que ter janela
C disponivel e pacotes pendentes, se nao cumprir estas condicoes sao tentados
C sequencialmente, todos os canais. Se o pacote recebido da Lan for de
C tamanho superior ao tamanho maximo permitido no correspondente canal na
C Psdn e' feita a sua segmentacao de modo a ser respeitado esse limite.
C Cada pacote recebido (ou gerado por segmentacao) ocupa um ou mais
C buffers consoante o seu tamanho. A recepcao de um pacote faz avancar a
C janela. E' registado o tempo de chegada dos pacotes ao Relay para,
C aquando da sua transmissao, ser calculado o atraso sofrido.
C Sempre que faltarem buffers para guardar os pacotes recebidos e'
C incrementada a variavel NBufFalta.
C

```

```

BEGIN RecepLan
TIME OF TxRxLan LE 0
CanRxLan = RANDOM (NCanal, Seed)
FOR J = Canallan
  TIME OF TxRxLan LE 0
  I = MODULO (CanRxLan + (J - 1), NCanal) + 1
  JanOcLan OF Canallan I LT Klan OF Canallan I
  FIND FIRST PacLan A IN WaitLan OF Canallan I
  ADD FIX(0.001 * (CLOCK - TInLan OF PacLan A)) TO
*      AWaitLan OF Canallan I
  Duracao = FIX ((CPLan OF PacLan A) * 8 / TaxaLan)
  TIME OF TxRxLan = Duracao
  P = 0
  CHAIN
  CPLan OF PacLan A GT MxPacPsdn OF CanalPsdn I
  N = CPLan OF PacLan A / MxPacPsdn OF CanalPsdn I
  CPLan OF PacLan A = MODULO (CPLan OF PacLan A,
*      MxPacPsdn OF CanalPsdn I)
  FOR 1 TO N
  CHAIN
    FIND FIRST PacLan C IN PoolLan
    TInReLan OF PacLan C = CLOCK + Duracao
    MbitLan OF PacLan C = 1
    CPLan OF PacLan C = MxPacPsdn OF CanalPsdn I
    PacLan C FROM PoolLan INTO WRelLan
*      OF Canallan I AFTER Duracao
    P = P + (CPLan OF PacLan C / SizeBuf)
    CHAIN
      MODULO (CPLan OF PacLan C, SizeBuf) NE 0
      P = P + 1
      OR CONTINUE
      OR TYPE "Numero insuficiente de pacotes em PoolLan"
*      AND FINISH
    OR CONTINUE
  TInReLan OF PacLan A = CLOCK + Duracao
  OcuRxLan OF Canallan I + Duracao AFTER Duracao
  LastTxRx = MAXOF (LastTxRx, CLOCK + Duracao, 2)
  JanOcLan OF Canallan I + 1 AFTER Duracao
  PacRxLan OF Canallan I + 1 AFTER Duracao
  PacLan A FROM WaitLan OF Canallan I INTO WRelLan OF Canallan I
*      AFTER Duracao
  N = CPLan OF PacLan A / SizeBuf
  CHAIN
    MODULO (CPLan OF PacLan A, SizeBuf) NE 0
    N = N + 1
    OR CONTINUE
  FOR 1 TO (N + P)
  CHAIN
    FIND FIRST Buffer B IN BufLivre
    Buffer B FROM BufLivre INTO BufOcupado
    NBufLan OF Canallan I + 1
    OR NBufFalta = NBufFalta + 1
  RecepLan + 1 AFTER Duracao

```

```

C-----
C      Actividade: Transmissao PSDN (TransPsdn)
C
C Esta actividade simula a transmissao pelo Relay de pacotes nos canais
C da Psdn. Os pacotes sao reassemblados sempre que a sequencia Mbit dos
C pacotes pendentes o permita. A reassemblagem e' feita respeitando o
C limite fixado para o tamanho maximo de pacote, de cada canal. Os canais
C sao servidos de acordo com o seu Indice de Partilha (IndPart). Depois de
C transmitido o pacote os Indices de Partilha de todos os canais sao
C actualizados.
C

```

```

BEGIN TransPsdn
TIME OF TxPsdn LE 0
MinIndPart = 9999999
Canal = 1
FOR I = CanalLan
  CHAIN
    EXISTS PacLan IN WRelLan OF Canallan I
    IndPrLan OF Canallan I LE MinIndPart
    MinIndPart = IndPrLan OF Canallan I
    Canal = I
    OR CONTINUE
  I = Canal
  EXISTS PacLan IN WRelLan OF Canallan I
  PacTx = 0
  L = 0
  Falhou = 0
  FOR PacLan IN WRelLan OF Canallan I
    CHAIN
      Falhou EQ 0
      MbitLan OF PacLan = 1
      L + CPLan OF PacLan LE MxPacPsdn OF CanalPsdn I
      PacTx = PacTx + 1
      L = L + CPLan OF PacLan
      OR Falhou = 1
    Duracao = FIX (L * 8 / TaxaPsdn)
    TIME OF TxPsdn = Duracao
    OcuTxPsdn OF Canallan I + Duracao AFTER Duracao
    LastTxRx = MAXOF (LastTxRx, CLOCK + Duracao, 2)
    PacTxPsdn OF Canallan I + 1 AFTER Duracao
    FOR 1 TO PacTx
      FIND FIRST PacLan A IN WRelLan OF Canallan I
      ADD FIX(0.0001 * (CLOCK - TInReLan OF PacLan A)) TO
      *                               AWRelLan OF Canallan I
      PacLan A FROM WRelLan OF Canallan I INTO PoolLan
      *                               AFTER Duracao
      N = CPLan OF PacLan A / SizeBuf
      CHAIN
        MODULO (CPLan OF PacLan A, SizeBuf) NE 0
        N = N + 1
        OR CONTINUE
      FOR 1 TO N
        NBufLan OF Canallan I GT 0
        FIND FIRST Buffer B IN BufOcupado
        Buffer B FROM BufOcupado INTO BufLivre AFTER Duracao
        NBufLan OF Canallan I - 1
        TransPsdn + 1 AFTER Duracao
      IndPrLan OF Canallan I = FIX (1000.0 *
      * ( ( (LanLastCalIndP * 0.001 * IndPrLan OF Canallan I) +
      * (L * 8.0 / (0.000001 * QoStrLan OF Canallan I)) ) /
      * (1. * (CLOCK + Duracao)) ) )
      FOR J = Canallan
        CHAIN
          J NE I
          IndPrLan OF Canallan J = FIX (1000 * (LanLastCalIndP *
          * 0.001 * IndPrLan OF Canallan J) /
          * (1. * (CLOCK + Duracao)))
          OR CONTINUE
        LanLastCalIndP = CLOCK + Duracao

```



```

C-----
C      Actividade: Confirmacao LAN (ConfLan)
C
C Esta actividade simula o envio de confirmacoes pelo Relay 'as estacoes
C na Lan. As confirmacoes sao simuladas pelo destravamento da janela
C dos canais. O destravamento so' e' feito se existir espaco para a
C recepcao de pelo menos um pacote nos buffers reservados a um canal.
C O destravamento e' feito com um atraso composto de uma parte constante
C e de uma parte obtida de uma cadencia de Poisson para simular o atraso
C na transmissao e propagacao das confirmacoes. As confirmacoes sao feitas
C rotativamente em todos os canais da Lan.
C
      BEGIN ConfLan
      FOR I = Canallan
        JanOcLan OF Canallan I GT 0
        BExist = MxBufLan OF Canallan I - NBufLan OF Canallan I
        BPac = MxPacLan OF Canallan I / SizeBuf
        BRes = (Klan OF Canallan I - JanOcLan OF Canallan I) * BPac
        BLivre = BExist - BRes
        J = BLivre / BPac
        CHAIN
          J GT 0
          Duracao = TConfLan + NEGEXP (TVarLanConf, Seed)
          JanOcLan OF Canallan I + (- J) AFTER Duracao
          ConfLan + J AFTER Duracao

```



```

C*****
C      Seccao de Finalizacao

      FINALISATION

      TYPE / **
      TYPE '@@@@@@@@@@@@ RELATORIO FINAL DA SIMULACAO @@@@@@@@@@@@@//

      TYPE '##### Dados da simulacao' /

      TYPE 'Numero de actividades:            ' NumAct
      TYPE 'Granularidade da simulacao (u seg):      1'
      TYPE 'Tempo de simulacao (seg):                '+4 (1.* NumAct/1000000)
      TYPE 'Ocorrencia da ultima Tx/Rx:             ' LastTxRx, /

      TYPE 'Numero de canais com a LAN e com a PSDN, no RELAY:  '*4
      *      NCanal /

      TYPE 'Capacidade maxima de armazenamento do relay (buffers): '*4
      *      NBuffer
      TYPE 'Capacidade de armazenamento de cada buffer (bytes): '*4
      *      SizeBuf
      TYPE 'Minimo recomendado de bufers para esta configuracao: '*4
      *      (NeedBuf / SizeBuf)
      TYPE 'Numero de bufers necessarios aos parametros QoS:     '*4
      *      (SomaBuf)

      TYPE / '>>>>>>>>>>> Relativos a LAN' /

      TYPE 'Factor de ocupacao do meio por outras estacoes da LAN '*5,+2
      *      (FactOcupMeio * 100) ' %'
      TYPE 'Tempo medio de ocupacao do meio em cada transmissao '*5,
      *      MediaOcmFis, ' us'
      TYPE 'Comprimento minimo dos quadros na LAN                    '*5,
      *      MinFraLan, ' bytes' /

      TYPE 'Debito maximo da LAN (bits/seg):                '
      *      FIX(TaxaLan * 1000000)
      TYPE 'Tempo de confirmacao na LAN (u seg):              ' TConfLan
      TYPE 'Limite das filas dos canais na LAN (pacotes):     ' LiBufLan

      TYPE / '>>>>>>>>>>> Relativos a PSDN' /

      TYPE 'Debito maximo da PSDN (bits/seg):                '
      *      FIX(TaxaPsdn*1000000)
      TYPE 'Tempo de confirmacao na PSDN (u seg):              ' TConfPsdn
      TYPE 'Limite das filas dos canais na PSDN (pacotes):    ' LiBufLan

      TYPE // '##### Resultados da simulacao' /

      TYPE '>>>>>>>>>>>>>>>>> Cargas medias  ' /

      CgLanPsdn = 0.0
      FOR I = Canallan
      CgLanPsdn = CgLanPsdn +
      * (8 * CMDLan OF Canallan I / (1.0 * IMCLan OF Canallan I))
      TYPE 'Carga media LAN -> PSDN (rel. a capac. LAN): '+6
      *      (CgLanPsdn / (1000 * TaxaLan))
      TYPE 'Carga media LAN -> PSDN (rel. a capac. PSDN): '+6
      *      (CgLanPsdn / (1000 * TaxaPsdn)), /

      CgPsdnLan = 0.0
      FOR I = CanalPsdn
      CgPsdnLan = CgPsdnLan +
      * (8 * CMDPsdn OF CanalPsdn I / (1.0 * IMCPsdn OF CanalPsdn I))
      TYPE 'Carga media PSDN -> LAN (rel. a capac. LAN): '+6
      *      (CgPsdnLan / (1000 * TaxaLan))
      TYPE 'Carga media PSDN -> LAN (rel. a capac. PSDN): '+6
      *      (CgPsdnLan / (1000 * TaxaPsdn)), /

      TYPE '>>>>>>>>>>>>>>>>> Debitos medios (rel. a ultima Tx/Rx)' /

```

```
TYPE 'Debito medio total na LAN (rel. ao deb. max.): '+6
*      (1-ATxRxLan/(1.*CLOCK))
Debito = 0
FOR I = Canallan
  Debito = Debito + OcuRxLan OF Canallan I
TYPE 'Debito medio LAN -> Relay (rel. ao deb. max.): '+6
*      (Debito / (1.*LastTxRx))
Debito = 0
FOR I = CanalPsdn
  Debito = Debito + OcuTxLan OF CanalPsdn I
TYPE 'Debito medio Relay -> LAN (rel. ao deb. max.): '+6
*      (Debito / (1.*LastTxRx)), /

Debito = 0
FOR I = CanalPsdn
  Debito = Debito + OcuRxPsdn OF CanalPsdn I
TYPE 'Debito medio PSDN -> Relay (rel. ao deb. max.): '+6
*      (Debito / (1.*LastTxRx))
Debito = 0
FOR I = Canallan
  Debito = Debito + OcuTxPsdn OF Canallan I
TYPE 'Debito medio Relay -> PSDN (rel. ao deb. max.): '+6
*      (Debito / (1.*LastTxRx)), /

TYPE '>>>>>>>>>>>>>>> Buffers do RELAY' /

TYPE 'Ocupacao media dos Buffers do RELAY: '+2
*      (MEAN (TBufOcupado))
TYPE 'Ocorrencias de falta de Buffers no Relay: ' NBufFalta, /

TYPE / '>>>>>>>>>>>>>>> Transito de pacotes e N-DATA.req' /

TYPE 'Numero de transmissoes na Lan pelas outras estacoes: '
*      MFisOcupado, /
TYPE 'Numero de N-DATA.req geradas na LAN destinados ao RELAY: '
*      ChLan
TYPE 'Numero de pacotes gerados na LAN destinados ao RELAY: '
*      ChPacLan
TYPE 'Numero de pacotes recebidos pelo RELAY, vindos da LAN: '
*      RecepLan
TYPE 'Numero de pacotes confirmados pelo RELAY, vindos da LAN: '
*      ConfLan
TYPE 'Numero de pacotes transmitidos pelo RELAY, na PSDN: '
*      TransPsdn, /

TYPE 'Numero de N-DATA.req geradas na PSDN destinados ao RELAY: '
*      ChPsdn
TYPE 'Numero de pacotes gerados na PSDN destinados ao RELAY: '
*      ChPacPsdn
TYPE 'Numero de pacotes recebidos pelo RELAY, vindos da PSDN: '
*      RecepPsdn
TYPE 'Numero de pacotes confirmados pelo RELAY, vindos da PSDN: '
*      ConfPsdn
TYPE 'Numero de pacotes transmitidos pelo RELAY, na LAN : '
*      TransLan, /

TYPE '>>>>>>>>>>>>>>> Transito LAN -> RELAY -> PSDN', /

TYPE '--', FOR 1 TO NCanal + 2 ('-----')
TYPE '!Compr !', FOR I = 1 TO NCanal (*6 I, '!'), ' Media !'
TYPE '--', FOR 1 TO NCanal + 2 ('-----')

Media = 0
FOR I = Canallan
  Media = Media + CMDLan OF Canallan I
Media = 1.0 * Media / NCanal
TYPE '!Compr !', FOR 1 TO NCanal (' !'), ' !'
TYPE '!Medio !', FOR I = Canallan (*6 CMDLan OF Canallan I, '!')
*      ,*7 Media, '! '
TYPE '!N-DATA!', FOR 1 TO NCanal (' !'), ' !'
TYPE '--', FOR 1 TO NCanal + 2 ('-----')
```

```

Media = 0
FOR I = Canallan
  Media = Media + IMCLan OF Canallan I
Media = 1.0 * Media / NCanal
TYPE '!Interv!', FOR 1 TO NCanal (' '), '!'
TYPE '!Cheg !', FOR I = Canallan (*6 IMCLan OF Canallan I, '!')
*
  ,*7 Media, '!'
TYPE '!N-DATA!', FOR 1 TO NCanal (' '), '!'
TYPE '--', FOR 1 TO NCanal + 2 ('-----')

Media = 0
FOR I = Canallan
  Media = Media + (NDLan OF Canallan I - CNDLan OF Canallan I)
Media = 1.0 * Media / NCanal
TYPE '!Quant !', FOR 1 TO NCanal (' '), '!'
TYPE '!N-DATA!', FOR I = Canallan (*6 (NDLan OF Canallan I -
*
  CNDLan OF Canallan I, '!'), *7 Media, '!'
TYPE '!Gerado!', FOR 1 TO NCanal (' '), '!'
TYPE '--', FOR 1 TO NCanal + 2 ('-----')

Media = 0
FOR I = Canallan
  Media = Media + KLAN OF Canallan I
Media = 1.0 * Media / NCanal
TYPE '! ', FOR 1 TO NCanal (' '), '!'
TYPE '!Janela!', FOR I = Canallan (*6 KLAN OF Canallan I, '!')
*
  ,*7 Media, '!'
TYPE '! ', FOR 1 TO NCanal (' '), '!'
TYPE '--', FOR 1 TO NCanal + 2 ('-----')

Media = 0
FOR I = Canallan
  Media = Media + MxPacLan OF Canallan I
Media = 1.0 * Media / NCanal
TYPE '!Tamanh!', FOR 1 TO NCanal (' '), '!'
TYPE '!Maximo!', FOR I = Canallan (*6 MxPacLan OF Canallan I, '!')
*
  ,*7 Media, '!'
TYPE '!Pacote!', FOR 1 TO NCanal (' '), '!'
TYPE '--', FOR 1 TO NCanal + 2 ('-----')

Media = 0
FOR I = Canallan
  Media = Media + MxBufLan OF Canallan I
Media = 1.0 * Media / NCanal
TYPE '!Quant !', FOR 1 TO NCanal (' '), '!'
TYPE '!Buffs !', FOR I = Canallan (*6 MxBufLan OF Canallan I, '!')
*
  ,*7 Media, '!'
TYPE '!Atrib !', FOR 1 TO NCanal (' '), '!'
TYPE '--', FOR 1 TO NCanal + 2 ('-----')

Media = 0
FOR I = Canallan
  Media = Media + PacRxLan OF Canallan I
Media = 1.0 * Media / NCanal
TYPE '!Pacots!', FOR 1 TO NCanal (' '), '!'
TYPE '!Receb !', FOR I = Canallan (*6 PacRxLan OF Canallan I, '!')
*
  ,*7 Media, '!'
TYPE '!Lan !', FOR 1 TO NCanal (' '), '!'
TYPE '--', FOR 1 TO NCanal + 2 ('-----')

Media = 0
FOR I = Canallan
  Media = Media + OcuRxLan OF Canallan I
Media = (1.0 * Media / NCanal) / LastTxRx
TYPE '!Debito!', FOR 1 TO NCanal (' '), '!'
TYPE '!Lan > !', FOR I = Canallan (*6 (OcuRxLan OF Canallan I /
*
  (1. * LastTxRx)), '!'), *7 Media, '!'
TYPE '!Relay !', FOR 1 TO NCanal (' '), '!'
TYPE '--', FOR 1 TO NCanal + 2 ('-----')

Media = 0

```



```

*          CanalPsdn I, '!') ,*7 Media, '!
TYPE '!N-DATA!', FOR 1 TO NCanal ('      !'), '      !'
TYPE '--', FOR 1 TO NCanal + 2 ('-----')

Media = 0
FOR I = CanalPsdn
  Media = Media + IMCPsdn OF CanalPsdn I
Media = 1.0 * Media / NCanal
TYPE '!Interv!', FOR 1 TO NCanal ('      !'), '      !'
TYPE '!Cheg  !', FOR I = CanalPsdn (*6 IMCPsdn OF
*          CanalPsdn I, '!') ,*7 Media, '!
TYPE '!N-DATA!', FOR 1 TO NCanal ('      !'), '      !'
TYPE '--', FOR 1 TO NCanal + 2 ('-----')

Media = 0
FOR I = CanalPsdn
  Media = Media + (NDPsdn OF CanalPsdn I -
*          CNDPsdn OF CanalPsdn I)
Media = 1.0 * Media / NCanal
TYPE '!Quant !', FOR 1 TO NCanal ('      !'), '      !'
TYPE '!N-DATA!', FOR I = CanalPsdn (*6 (NDPsdn OF
*          CanalPsdn I - CNDPsdn OF CanalPsdn I), '!') ,*7 Media, '!
TYPE '!Gerado!', FOR 1 TO NCanal ('      !'), '      !'
TYPE '--', FOR 1 TO NCanal + 2 ('-----')

Media = 0
FOR I = CanalPsdn
  Media = Media + KPsdn OF CanalPsdn I
Media = 1.0 * Media / NCanal
TYPE '!      !', FOR 1 TO NCanal ('      !'), '      !'
TYPE '!Janela!', FOR I = CanalPsdn (*6 KPsdn OF
*          CanalPsdn I, '!') ,*7 Media, '!
TYPE '!      !', FOR 1 TO NCanal ('      !'), '      !'
TYPE '--', FOR 1 TO NCanal + 2 ('-----')

Media = 0
FOR I = CanalPsdn
  Media = Media + MxPacPsdn OF CanalPsdn I
Media = 1.0 * Media / NCanal
TYPE '!Tamanh!', FOR 1 TO NCanal ('      !'), '      !'
TYPE '!Maximo!', FOR I = CanalPsdn (*6 MxPacPsdn OF
*          CanalPsdn I, '!') ,*7 Media, '!
TYPE '!Pacote!', FOR 1 TO NCanal ('      !'), '      !'
TYPE '--', FOR 1 TO NCanal + 2 ('-----')

Media = 0
FOR I = CanalPsdn
  Media = Media + MxBufPsdn OF CanalPsdn I
Media = 1.0 * Media / NCanal
TYPE '!Quant !', FOR 1 TO NCanal ('      !'), '      !'
TYPE '!Bufs !', FOR I = CanalPsdn (*6 MxBufPsdn OF
*          CanalPsdn I, '!') ,*7 Media, '!
TYPE '!Atrib !', FOR 1 TO NCanal ('      !'), '      !'
TYPE '--', FOR 1 TO NCanal + 2 ('-----')

Media = 0
FOR I = CanalPsdn
  Media = Media + PacRxPsdn OF CanalPsdn I
Media = 1.0 * Media / NCanal
TYPE '!Pacots!', FOR 1 TO NCanal ('      !'), '      !'
TYPE '!Receb !', FOR I = CanalPsdn (*6 PacRxPsdn OF
*          CanalPsdn I, '!') ,*7 Media, '!
TYPE '!Psdn !', FOR 1 TO NCanal ('      !'), '      !'
TYPE '--', FOR 1 TO NCanal + 2 ('-----')

Media = 0
FOR I = CanalPsdn
  Media = Media + OcuRxPsdn OF CanalPsdn I
Media = (1.0 * Media / NCanal) / LastTxRx
TYPE '!Debito!', FOR 1 TO NCanal ('      !'), '      !'
TYPE '!Psdn >!', FOR I = CanalPsdn (*6 (OcuRxPsdn OF CanalPsdn I /
*          (1. * LastTxRx)), '!') ,*7 Media, '!

```



```
TYPE 'Histograma dos atrasos na fila na LAN:'
TYPE PICTURE (AWaitLan OF CanalLan I)
TYPE 'Histograma do comprimento da fila na LAN: '
TYPE PICTURE (TWaitLan OF CanalLan I)
TYPE 'Histograma dos atrasos na fila no Relay:'
TYPE PICTURE (AWRelLan OF CanalLan I)
TYPE 'Histograma do comprimento da fila no RELAY:'
TYPE PICTURE (TWRelLan OF CanalLan I)

TYPE /'>>>>>>>>>> Histogramas relativos aos canais da PSDN' /

FOR I = CanalPsdn
TYPE '>>>> Canal ' *2 I, ' da PSDN' /
TYPE 'Histograma dos atrasos na fila na PSDN:'
TYPE PICTURE (AWaitPsdn OF CanalPsdn I)
TYPE 'Histograma do comprimento da fila na PSDN:'
TYPE PICTURE (TWaitPsdn OF CanalPsdn I)
TYPE 'Histograma dos atrasos na fila no Relay:'
TYPE PICTURE (AWRelPsdn OF CanalPsdn I)
TYPE 'Histograma do comprimento da fila no RELAY:'
TYPE PICTURE (TWRelPsdn OF CanalPsdn I)

TYPE /'>>>>>>>>>> Histogramas relativos aos Buffers do RELAY'/

TYPE 'Histograma da ocupacao dos Buffers do RELAY'
TYPE PICTURE(TBufOcupado)

OR CONTINUE

TYPE ////
```

```

C*****
C      Seccao de Dados
C
C Nota:
C Os tempos TConfLan, TConfPsdn, TVarLanConf e TVarLanConf sao
C arredondados e em microsegundos.
C Os tempos IMCLan, IMCPsdn, QoSdLlan e QoSdLPsdn sao em milisegundos.
C Taxas de transmissao em bits por microsegundo (TaxaLan e TaxaPsdn).
C Os comprimentos sao em bytes

```

## DATA

```

NumAct      40000000
NCanal       5
NBuffer      80
SizeBuf      128
Histo        0
Tick         0
Seed         12345

```

```

TaxaLan      10.0
TConfLan     1
TVarLanConf  2
LiBufLan     5
MinFraLan   64

```

```

MediaOcmFis  13
FactOcupMeio 0.001

```

```

TaxaPsdn     0.064000
TConfPsdn    20
TVarPsdnConf 20
LiBufPsdn    5

```

```

KLan         5      5      5      5      5
MxPacLan     512    512    512    512    512
CMDLan       1024   1024   1024   1024   1024
NDLan        1000   1000   1000   1000   1000
IMCLan       250    250    250    250    250
HopCtLan     1      1      1      1      1
QoSTrLan     12000  12000  12000  12000  12000
QoSdLlan     250    500    750    1000   1250

```

```

KPsdn        5      5      5      5      5
MxPacPsdn    128    128    128    128    128
CMDPsdn      1024   1024   1024   1024   1024
NDPsdn       1000   1000   1000   1000   100
IMCPsdn      250    250    250    250    250
HopCtPsdn    1      1      1      1      1
QoSTrPsdn    12000  12000  12000  12000  12000
QoSdLPsdn    250    500    750    1000   1250

```

END